The University of Sydney
School of Mathematics and Statistics

Solutions to Groups in Magma

MagmaMondays: 16 October 2023 Semester 2, 2023

Web Page: https://sites.google.com/view/magma-mondays/
Lecturer: Don Taylor

1. Suppose that X is an invertible 2×2 matrix over the finite field F of 11 elements. The function $\theta_X : M \mapsto X^{-1}MX$ is a linear transformation of the vector space of all 2×2 matrices over F. Furthermore θ is a homomorphism from the general linear group $\mathrm{GL}(2, F)$ to $\mathrm{GL}(4, F)$.

(a) Let F := GaloisField(11) and write a Magma function that returns the matrix of X with respect to the 'standard basis' of the vector space KMatrixSpace(F, 2, 2).

Solution:

```
F := GF(11);
V := KMatrixSpace(F,2,2);
B := Basis(V);
φ := func< X | Matrix(F,4,4,[Coordinates(V,X^-1*b*X) : b in B]) >;
```

(b) Find the image of the generators of GL(2, F) under the homomorphism θ and thereby find the order of the images of GL(2, F) and SL(2, F) in GL(4, F).

Solution:

```
imG := sub< GL(4,F) | [ φ(g) : g in Generators(GL(2,F)) ] >;
imS := sub< GL(4,F) | [ φ(g) : g in Generators(SL(2,F)) ] >;
#GL(2,F), #imG;
#SL(2,F), #imS;
```

The function ϕ can be turned into a homomorphism as follows.

```
θ := hom< GL(2,F) -> GL(4,F) | X :-> φ(X) >;
Kernel(θ);
Image(θ) eq imG;
```

2. Let σ_1, σ_2 and σ_3 be the Pauli matrices defined over the Gaussian field $\mathbb{Q}[i]$.

```
K<i> := QuadraticField(-1);
σ1 := Matrix(K,[[0,1], [1, 0]]);
σ2 := Matrix(K,[[0,i], [-i, 0]]);
σ3 := Matrix(K,[[1,0], [0,-1]]);
```

and put

```
θ := Matrix(K,[[i,0], [0,i]]);
```

Let E be the subgroup of GL(2, K) generated by σ_1, σ_2, σ_3 and θ. Show that the matrices $\theta\sigma_1$, $\theta\sigma_2$, $\theta\sigma_3$ generate the quaternion group Q and E is the central product of a cyclic group of order 4 and Q.

Solution:

```
G := GL(2,K);
E := sub< G | σ1, σ2, σ3, θ >;
```



```
Q := sub< G | [θ*g : g in [σ1, σ2, σ3]] >;
```

The following values

```
(Q.1)^2, (Q.2)^2, (Q.3)^2, Q.1*Q.2*Q.3;
```

are all equal to $-I$, where I is the identity. Thus Q is the quaternion group.

```
C := sub<G | θ >;
E eq sub< G | Q, C>;
C subset Centre(E);
#(C meet Q) eq 2;
```

3. Let *fano* be the 7-point plane, and as in the lecture, define a graph (call it Gr_1) on the points and lines by joining each line to the points not on it.

(a) Use Magma to show that the automorphism group of Gr_1 is isomorphic to the projective linear group PGL(2, 7).

Solution:

```
fano := FiniteProjectivePlane(2);
P := Points(fano);
L := Lines(fano);
vertices1 := {@<-1,i> : i in [1..7]@} join {@<-2,j> : j in [1..7]@};
edges1 := { { ← 1,i>, ← 2,j>} : i,j in [1..7] | P[i] notin L[j] };
Gr1 := Graph< vertices1 | edges1 >;
M1 := AutomorphismGroup(Gr1);
IsIsomorphic(PGL(2,7), M1);
```

```
true
```

(b) Let

```
P2 := {1..7};
L2 := {{1 + n, 1 + (n+1) mod 7, 1 + (n+3) mod 7} : n in [0..6]};
```

Define a graph Γ_2 by joining each triple X in L_2 to the points in its complement in P_2. Use Magma to show that Gr_1 is isomorphic to Γ_2.

Solution:

```
L2 := SetToSequence(L2);
E2 := &join[{ {i,7+k} : i in (P2 diff ln) } : k -> ln in L2 ];
Gr2 := Graph< {1..14} | E2 >;
IsIsomorphic(Gr1, Gr2);
```

```
true
```

4. Let M_1 be the automorphism group of the graph Gr_1 of Exercise 3.

(a) Check that there are 28 involutions of M_1 not in its derived group D.

Solution:

```
D := DerivedGroup(M1);
#{ x : x in M1 | x notin D and Order(x) eq 2 };
```

```
28
```

(b) Check that the involutions form a single conjugacy class in M_1 and that each involution interchanges the orbits of D.

Solution:

```
T := { x : x in M1 | x notin D and Order(x) eq 2 };
t := Rep(T);
T eq t^M1;
```

```
true
```

```
OO := Orbits(D);
forall{ x: x in T | OO[1]^x eq OO[2] };
```

```
true
```

(c) Check that there are 28 symmetric matrices in $\mathrm{SL}(3, 2)$. Find a connection between these 28 matrices and the conjugacy class of 28 involutions in M_1.

Solution:

```
S := SL(3,2);
U := { g : g in S | g eq Transpose(g) };
#U;
```

```
28
```

A symmetric matrix $J \in \mathrm{SL}(3, 2)$ defines a symmetric bilinear form $(u, v) \mapsto uJv^\top$ on the vector space of dimension 3 over F_2. Equivalently, J corresponds to a polarity of the projective plane, interchanging points and lines; i.e., an element of order 2 in the automorphism group of $\mathrm{SL}(3, 2)$. Using the points P and lines L from exercise 3

```
V := VectorSpace(GF(2),3);
ψ := function(J)
    inv := One(Sym(14));
    for i := 1 to #P do
        inv *:= Sym(14) ! (i, 7+Index(L, L ! Eltseq(V ! Eltseq(P[i]) * J)));
    end for;
    return inv;
end function;

{ψ(J) : J in U } eq T;
```

```
true
```

(d) The *stabiliser* in M_1 of a vertex v in the graph Gr_1 is the subgroup

```
H := Stabilizer(M1,1);
```

Find the orbits of the stabiliser on the vertices of the graph.

Solution:

```
OO := Orbits(H); OO;
```

```
[
 GSet{@ 1 @},
 GSet{@ 9, 12, 10 @},
 GSet{@ 8, 14, 11, 13 @},
 GSet{@ 2, 3, 5, 6, 7, 4 @}
]
```

(e) By exploring the action of H on its orbits (or otherwise) show that H is isomorphic to Sym(4).
(Hint: OrbitAction(H, orb), returns f, H_1, K, where f is a homomorphism from H to the group H_1 defined by the action of H on orb, and K is the kernel of f.)

Solution:

```
f, H1, K := OrbitAction(H, OO[3]);
H1;
```

```
Permutation group H1 acting on a set of cardinality 4
Order = 24 = 2^3 * 3
    (2, 3)
    (1, 4)
    (2, 4)
```

```
K;
```

```
Permutation group K acting on a set of cardinality 14
Order = 1
```

The kernel of the action of H on the orbit of length 4 is 1; i.e., H acts faithfully on this orbit. The order of H is 24 and therefore it is the group of all permutations of the 4 elements of the orbit.

5. Let Gr_2 be the graph on 36 vertices defined in the lecture. For this exercise you will need to hunt through the Magma Handbook to find out how to construct a semidirect product and a Chevalley group of type G_2.

**(a) Show that the automorphism group of Gr_2 is isomorphic to the group SU(3, 3) of 3×3 unitary matrices (with coefficients in the field $\mathbb{F}_9$ of 9 elements) extended by the field automorphism $\sigma : \mathbb{F}_9 \to \mathbb{F}_9 : x \mapsto x^3$.

Solution: We build on the code from Exercise 3.

```
F := [ <i,j> : i,j in [1..7] | P[i] in L[j] ];
vertices2 := {@ <0,0> @} join vertices1
      join {@ <i,j> : i,j in [1..7] | P[i] in L[j] @};
edges2 := {{<0,0>, ←1,i>} : i in [1..7] }
  join { {<0,0>,  ←2,i>} : i in [1..7] } join edges1
  join { {←1,i>,<j,k>} : i,j,k in [1..7] | P[i] in L[k] and P[j] in L[k] }
  join { {←2,i>,<j,k>} : i,j,k in [1..7] | P[j] in L[k] and P[j] in L[i] }
  join { {f,g} : f, g in F | f[1] ne g[1] and f[2] ne g[2]
            and (P[f[1]] in L[g[2]] or P[g[1]] in L[f[2]]) };
Gr2 := Graph< vertices2 | edges2 >;
M2 := AutomorphismGroup(Gr2);
```

Now construct the semidirect product of SU(3, 3) by the field automorphism.

```
S := SU(3,3);
A := AutomorphismGroup(S);
f := hom< C  ->  A | hom<S -> S | x  |->  FrobeniusImage(x,1) > >;
G := SemidirectProduct(S,C,f);
check, _ := IsIsomorphic(G,M2); check;
```

```
true
```

*(b) Show that the automorphism group of the graph Gr_2 is isomorphic to the group of Lie type $G_2(2)$.

Solution:

```
check := IsIsomorphic(M2, ChevalleyGroup("G", 2, 2)); check;
```

```
true
```

6. Check Janko's conditions for the derived group of the automorphism group of the Wales graph on 100 vertices (defined in the lecture). That is, the centre of a Sylow 2-subgroup is cyclic and the centraliser C of a central involution has a normal subgroup E such that $C/E \simeq \mathrm{Alt}(5)$.

(Hint. You can use the MAGMA intrinsics SylowSubgroup, Centre, Centraliser, pCore and **quo**<C|E>. Use the on-line Handbook at

`http://magma.maths.usyd.edu.au/magma/handbook/`

to find out how these commands work.)

Solution: The Wales graph can be constructed using the code from Exercises 3, 5 and the following.

```
edges := { {Index(vertices2, x) : x in edge} : edge in edges2 };
exists(t){ c[3] : c in Classes(M2) | c[1] eq 2 and c[2] eq 63 };
edges := { {Index(vertices2, x) : x in edge} : edge in edges2 };
exists(t){ c[3] : c in Classes(M2) | c[1] eq 2 and c[2] eq 63 };
X := SetSeq(Conjugates(M2, t));
edges join:= {{i, j+36} : i in [1..36], j in [1..63] | i^X[j] eq i};
edges join:= {{i+36, j+36} : i, j in [1..63] | Order(X[i]*X[j]) eq 4};
edges join:= { {i, 100} : i in [1..36] };
WalesGraph := Graph< 100 | edges >;
JJ2 := AutomorphismGroup(WalesGraph);
J2 := DerivedGroup(JJ2);
S2 := SylowSubgroup(J2, 2);
Z := Centre(S2);
#Z;
```

```
2
```

```
C := Centraliser(J2, Z.1);
E := pCore(C, 2);
#E;
```

```
32
```

```
check := IsIsomorphic(quo<C|E>, Alt(5)); check;
```

```
true
```

7. Factorise the group determinants of the five groups of order 12. (You can get the groups from the Small Groups Database.)

Warning. This can take rather a long time. Are there faster ways to factorise the group determinant?

Solution:

```
groupDet := function(G)
    n := #G;
    P := PolynomialRing(Integers(), n : Global);
    AssignNames(~P, ["x" cat IntegerToString(i) : i in [1..n]]);
    L := SetSeq(Set(G)); L := [h*g : g in L] where h is L[1]^-1;
    M := ZeroMatrix(P, n, n);
    for i -> x in L, j -> y in L do
        k := Index(L, x*y^-1);
        M[i,j] := P.k;
    end for;
    return M, Determinant(M);
end function;
for d := 1 to NumberOfSmallGroups(12) do
    "Group", d;
    G := SmallGroup(12, d);
    time M, D := groupDet(G);
    // time Factorisation(D);
end for;
```

8. Using MAGMA's cohomology intrinsics find all central extensions of Sym(5) by the cyclic group of order 2 and describe their structure.

Solution:

```
G := Sym(5);
CM := CohomologyModule(G, A) where A is TrivialModule(G, GF(2));
H2 := CohomologyGroup(CM, 2);
Dimension(H2);
```

```
2
```

```
extns := [ Extension(CM, v) : v in [H2| [0,0], [1,0], [0,1], [1,1]] ];
permgps := [ CosetImage(E, sub<E|>) : E in extns ];
[#DerivedGroup(X) : X in permgps ];
```

```
[ 60, 60, 120, 120 ]
```

```
[#Centre(X) : X in permgps ];
```

```
[ 2, 2, 2, 2 ]
```

```
[ Centre(X) subset DerivedGroup(X) : X in permgps ];
```

```
[ false, false, true, true ]
```

```
[exists{t : t in X | Order(t) eq 2 and t notin DerivedGroup(X)} : X in permgps];
```

```
[ true, true, true, false ]
```

```
check := IsIsomorphic(permgps[1], DirectProduct(CyclicGroup(2), G)); check;
```

```
true
```

Thus *permgps*[1] is the direct product $C_2 \times \mathrm{Sym}(5)$. It can be shown that *permgps*[2] is the semidirect product of Alt(5) by a cyclic group of order 4, where the element of order 4 acts on Alt(5) as an involution from Sym(5).